```
> restart;
> with(plots):
```

Height of the domain (y-coordinate)

```
> h:=1;
```

$$h := 1$$

Length of the domain (z-coordinate)

```
> L:=1;
```

$$L := 1$$

```
>
```

Governing Equation (Laplace's Equation)

```
> Eq1:=diff(phi(y,z),y$2)+diff(phi(y,z),z$2);
```

$$Eq1 := \left(\frac{\partial^2}{\partial y^2}\phi(y,z)\right)+\left(\frac{\partial^2}{\partial z^2}\phi(y,z)\right)$$

Boundary condition at y=0

```
> bcy1:=diff(phi(y,z),y);
```

$$bcy1 := \frac{\partial}{\partial y}\phi(y,z)$$

Boundary condition at y=1

```
> bcy2:=-phi(y,z);
```

$$bcy2 := -\phi(y,z)$$

Boundary condition at z=0

```
> bcz1:=diff(phi(y,z),z);
```

$$bcz1 := \frac{\partial}{\partial z}\phi(y,z)$$

Boundary condition at z=1

```
> bcz2:=-phi(y,z)+1;
```

$$bcz2 := -\phi(y,z)+1$$

Finite difference scheme to be used (2nd order-central difference for 2nd derivative)

```
> d2phidy2:=(phi[m+1,n]-2*phi[m,n]+phi[m-1,n])/dely^2;
  d2phidz2:=(phi[m,n+1]-2*phi[m,n]+phi[m,n-1])/delz^2;
```

$$d2phidy2 := \frac{\phi_{m+1,n}-2\,\phi_{m,n}+\phi_{m-1,n}}{dely^2}$$

$$d2phidz2 := \frac{\phi_{m,n+1}-2\,\phi_{m,n}+\phi_{m,n-1}}{delz^2}$$

Finite difference scheme to be used (2nd order-central difference for 1st derivative)

```
> dphidy:=(phi[m+1,n]-phi[m-1,n])/(2*dely);
> dphidz:=(phi[m,n+1]-phi[m,n-1])/(2*delz);
```

$$dphidy := \frac{1}{2}\,\frac{\phi_{m+1,n}-\phi_{m-1,n}}{dely}$$

$$dphidz := \frac{1}{2}\,\frac{\phi_{m,\,n+1} - \phi_{m,\,n-1}}{delz}$$

3 Point forward and backward differences for 1st derivative (to be applied at boundary conditions)

```
> dphidyf:=(-phi[2,n]+4*phi[1,n]-3*phi[0,n])/(2*dely);
  dphidyb:=(phi[Numy-1,n]-4*phi[Numy,n]+3*phi[Numy+1,n])/(2*delz);
  dphidzf:=(-phi[m,2]+4*phi[m,1]-3*phi[m,0])/(2*delz);
  dphidzb:=(phi[m,Numz-1]-4*phi[m,Numz]+3*phi[m,Numz+1])/(2*delz);
```

$$dphidyf := \frac{1}{2}\,\frac{-\phi_{2,\,n} + 4\,\phi_{1,\,n} - 3\,\phi_{0,\,n}}{dely}$$

$$dphidyb := \frac{1}{2}\,\frac{\phi_{Numy-1,\,n} - 4\,\phi_{Numy,\,n} + 3\,\phi_{Numy+1,\,n}}{delz}$$

$$dphidzf := \frac{1}{2}\,\frac{-\phi_{m,\,2} + 4\,\phi_{m,\,1} - 3\,\phi_{m,\,0}}{delz}$$

$$dphidzb := \frac{1}{2}\,\frac{\phi_{m,\,Numz-1} - 4\,\phi_{m,\,Numz} + 3\,\phi_{m,\,Numz+1}}{delz}$$

Number of interior node points in y

```
> Numy:=5;
```

$$Numy := 5$$

Number of interior node points in z

```
> Numz:=5;
```

$$Numz := 5$$

```
>
> dely:=(h)/(Numy+1);
```

$$dely := \frac{1}{6}$$

```
> delz:=(L)/(Numz+1);
```

$$delz := \frac{1}{6}$$

```
>
```

Develop finite difference equations for z boundary conditions for all y

```
> for j from 0 to Numy+1 do
  Eq[j,0]:=subs(diff(phi(y,z),z)=dphidzf,phi(y,z)=phi[j,0],m=j,bcz1)
  :
  Eq[j,Numz+1]:=subs(diff(phi(y,z),z)=dphidzb,phi(y,z)=phi[j,Numz+1]
  ,m=j,bcz2):
  od;
```

$$Eq_{0,\,0} := -3\,\phi_{0,\,2} + 12\,\phi_{0,\,1} - 9\,\phi_{0,\,0}$$

$$Eq_{0,\,6} := -\phi_{0,\,6} + 1$$

$$Eq_{1,\,0} := -3\,\phi_{1,\,2} + 12\,\phi_{1,\,1} - 9\,\phi_{1,\,0}$$

$$Eq_{1,6} := -\phi_{1,6} + 1$$

$$Eq_{2,0} := -3\,\phi_{2,2} + 12\,\phi_{2,1} - 9\,\phi_{2,0}$$

$$Eq_{2,6} := -\phi_{2,6} + 1$$

$$Eq_{3,0} := -3\,\phi_{3,2} + 12\,\phi_{3,1} - 9\,\phi_{3,0}$$

$$Eq_{3,6} := -\phi_{3,6} + 1$$

$$Eq_{4,0} := -3\,\phi_{4,2} + 12\,\phi_{4,1} - 9\,\phi_{4,0}$$

$$Eq_{4,6} := -\phi_{4,6} + 1$$

$$Eq_{5,0} := -3\,\phi_{5,2} + 12\,\phi_{5,1} - 9\,\phi_{5,0}$$

$$Eq_{5,6} := -\phi_{5,6} + 1$$

$$Eq_{6,0} := -3\,\phi_{6,2} + 12\,\phi_{6,1} - 9\,\phi_{6,0}$$

$$Eq_{6,6} := -\phi_{6,6} + 1$$

Develop finite difference equations for y boundary conditions for all z

```
> for j from 0 to Numz+1 do
  Eq[0,j]:=subs(diff(phi(y,z),y)=dphidyf,phi(y,z)=phi[0,n],n=j,bcy1)
  :
  Eq[Numy+1,j]:=subs(diff(phi(y,z),y)=dphidyb,phi(y,z)=phi[Numy+1,n]
  ,n=j,bcy2):
  od;
```

$$Eq_{0,0} := -3\,\phi_{2,0} + 12\,\phi_{1,0} - 9\,\phi_{0,0}$$

$$Eq_{6,0} := -\phi_{6,0}$$

$$Eq_{0,1} := -3\,\phi_{2,1} + 12\,\phi_{1,1} - 9\,\phi_{0,1}$$

$$Eq_{6,1} := -\phi_{6,1}$$

$$Eq_{0,2} := -3\,\phi_{2,2} + 12\,\phi_{1,2} - 9\,\phi_{0,2}$$

$$Eq_{6,2} := -\phi_{6,2}$$

$$Eq_{0,3} := -3\,\phi_{2,3} + 12\,\phi_{1,3} - 9\,\phi_{0,3}$$

$$Eq_{6,3} := -\phi_{6,3}$$

$$Eq_{0,4} := -3\,\phi_{2,4} + 12\,\phi_{1,4} - 9\,\phi_{0,4}$$

$$Eq_{6,4} := -\phi_{6,4}$$

$$Eq_{0,5} := -3\,\phi_{2,5} + 12\,\phi_{1,5} - 9\,\phi_{0,5}$$

$$Eq_{6,5} := -\phi_{6,5}$$

$$Eq_{0,6} := -3\,\phi_{2,6} + 12\,\phi_{1,6} - 9\,\phi_{0,6}$$

$$Eq_{6,6} := -\phi_{6,6}$$

```
> #printlevel:=2;
>
```

```
>
```

Develop finite difference equations for all interior points using the governing equation

```
> for j from 1 to Numy do
  for k1 from 1 to Numz do
  Eq[j,k1]:=subs(diff(phi(y,z),y$2)=d2phidy2,diff(phi(y,z),z$2)=d2ph
  idz2,diff(phi(y,z),y)=dphidy,diff(phi(y,z),z)=dphidz,phi(y,z)=phi[
  j,k1],n=k1,m=j,Eq1);
  od;od;
> #printlevel:=1;
```

Collect all equations into a single list

```
> eqs:=[seq(seq(Eq[p,q],p=0..Numy+1),q=0..Numz+1)]:
```

Collect all variables into a single list

```
> vars:=[seq(seq(phi[i,j],i=0..Numz+1),j=0..Numy+1)]:
>
```

Count number of equations

```
> n1:=nops(eqs);
```

$$n1 := 49$$

```
>
```

Convert all variables from the form phi[i,j] to YY[i]

```
> vars2:=[seq(vars[i]=YY[i](t),i=1..n1)]:
```

Perturbation parameter to be used

```
> mu:=1e-3:
```

Substitute new variables into the equations

```
> Eqs:=subs(vars2,eqs):
```

Eqs2 is the standard false transient formulation

```
> Eqs2:=seq(diff(YY[i](t),t)=Eqs[i],i=1..n1):
```

Eqs3 is the perturbation approach described in the paper

```
> Eqs3:=seq(mu*diff(Eqs[i],t)=-Eqs[i],i=1..n1):
>
```

This is an initial guess for all values of phi to be used in the IVP solver

```
> ics2:=seq(YY[i](0)=1,i=1..n1):
```

Solver the standard false transient formulation with Maple's dsolve

```
> temp:=time[real]():sol2a:=dsolve({Eqs2,ics2},type=numeric,implicit
  =true):time[real]()-temp;
```

$$0.343$$

Solver the perturbation formulation with Maple's dsolve

```
> temp:=time[real]():sol3a:=dsolve({Eqs3,ics2},type=numeric,implicit
  =true):time[real]()-temp;
```

$$0.343$$

```
>
>
```

Plot the convergence of the standard false transient (red) and the perturbation approach (green). YY[1] is phi at y=0, z=0

```
> t11:=time[real]():p2:=odeplot(sol2a,[t,YY[1](t)],0..10):t11-time[r
  eal]();
  t11:=time[real]():p3:=odeplot(sol3a,[t,YY[1](t)],0..10,color=green
  ):t11-time[real]();

  display(p2,p3);
```

-1.030

-80.216



Calculate converged value from false transient approach (value at pseudo time=50)

```
> s2:=sol2a(50):
```

Calculate converged value from perturbation approach (value at pseudo time=10)

```
> s3:=sol3a(10):
>
>
```

Convert the solution (from s3, as a list), as an array

```
> for p from 0 to Numy+1 do
> for q from 0 to Numz+1 do
  phi3[p,q]:=subs(vars2,s3,phi[p,q]);
  yt:=p/(Numy+1);
```

```
zt:=q/(Numz+1);
od;
od:
>
>
>
```

Plot the solution as found from the perturbation approach

```
> sol3:=matrix([seq([seq(phi3[p,q],p=0..Numy+1)],q=0..Numz+1)]):
> matrixplot(sol3,axes=boxed);
```




```
>
```